# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=http-client-restricted
pkgname=haskell-http-client-restricted
pkgver=0.0.5
pkgrel=98
pkgdesc="Restricting the servers that http-client will use"
url="https://hackage.haskell.org/package/http-client-restricted"
license=('MIT')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-connection' 'haskell-data-default' 'haskell-http-client'
         'haskell-http-client-tls' 'haskell-network' 'haskell-network-bsd' 'haskell-utf8-string')
makedepends=('ghc')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('d5876c02c1c6989f1da4741e191f18554efd448339e83bb9b267edec0e414141221c4f770d9f8e48c9d5f4a8fac9d58295e5f57e8c877710c95a71419e53cf3d')

build() {
    cd $_hkgname-$pkgver    
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
